AND8250/D

Zener Macro-Models Provide Accurate SPICE Simulations

Prepared by: Jim Lepkowski
ON Semiconductor

ON Semiconductor®

http://onsemi.com

APPLICATION NOTE

INTRODUCTION

Zener macro-models provide an accurate SPICE simulation of a diode's current versus voltage characteristics. The macro-models are created by combining standard SPICE devices into a sub-circuit. Zener macro-models offer several advantages over the standard diode model available in SPICE, including a more accurate representation of the breakdown characteristics.

Zener I vs. V Characteristics

The current versus voltage relationship of a Zener diode is shown in Figure 1. The four operational regions are the forward, leakage, breakdown and the maximum current regions. Zener diodes are typically used as voltage regulators; thus, the reverse bias breakdown region is the normal operating area for the device.



I_F = Forward current
V_F = Voltage at I_F
I_R = Reverse leakage current
V_R = Voltage at I_R
I_{ZK} = Test current for voltage V_{BR}
V_{BR} = Voltage at I_{ZK}
Z_{ZK} = Dynamic impedance at I_{ZK} $= \Delta V_Z / \Delta I_{ZK}$
I_{ZT} = Test current for voltage V_Z
V_Z = Voltage at current I_{ZT}
Z_{ZT} = Dynamic impedance at I_{ZT} $= \Delta V_Z / \Delta I_{ZT}$
I_{ZM} = Maximum DC steady-state current
V_{ZM} = Voltage at I_{ZM} (typically not defined on the data sheet)

Figure 1. Zener Diode I vs. V Characteristics and Specification Definitions

Zener Diode SPICE Model Options

The three basic methods used to create Zener diode SPICE models are listed below.

1. Curve Fit Models
2. SPICE Diode Statement
3. Macro-Model Subcircuit

Curve Fit Models

Curve fit SPICE models are an attractive modeling option in versions of SPICE that have analog behavior modeling statements. The mathematical functions and "If-Then-Else" commands can be used to perform conditional branching to accurately match the Zener's impedance in the four operating regions. Curve fit models are not as popular as the Diode 'D' statement or macro-model options because the analog behavioral statements are not available in some versions of SPICE. In contrast, the SPICE statements used in the diode and macro-model are compatible with almost all versions of SPICE.

Curve fit models are created using a polynomial expression in the SPICE voltage controlled current source 'G' command. The polynomials can be generated using the curve fit feature of a mathematical program such as MathCad or Excel to form a transfer function of the Zener's impedance. A negative feature of the curve fit approach is that this option creates a mathematical model, while in contrast the macro-model approach produces an equivalent circuit to represent the Zener.

SPICE Diode Statement

The majority of the Zener SPICE models available in the industry are created using the SPICE 'D' diode statement. Table 1 provides the variables available with the PSPICE 'D' diode model. An example of a 'D' statement is shown below.

```
*  anode  cathode   model name
D1   2        1         MD1
.MODEL MD1 D IS=6.57933e-10 N=1.84949 XTI=1
+RS=1.022CJO=1.5e-10 TT=1e-08
```

There are several restrictions that limit the accuracy of using the diode 'D' statement to model a Zener. First, the diode statement does not have a provision for defining a separate series resistance for the forward and reverse bias breakdown regions. The resistances in the two regions are not equal; thus, it is not possible to accurately model the slope of the voltage versus current characteristic in both regions. Next, the 'D' statement does not have a variable to model the variance of the Zener voltage with temperature (T). The polarity and magnitude of $\Delta V_Z/\Delta T$ is a function of the breakdown voltage. If V_Z is < 5.6 V, $\Delta V_Z/\Delta T \approx -2.0$ mV/°C. If V_Z is > than 5.6 V, the $\Delta V_Z/\Delta T$ temperature coefficient is positive and increases as a function of the magnitude of V_Z. Also, the diode statement does not have a variable to limit the current to a maximum value that matches the Zener's power dissipation capability.

Table 1. Default Values of the PSPICE Diode Statement 'D' Variables

Variable	Parameter	SPICE Default Value	ON Macro-Model Default Value	Units
IS	Saturation Current	1 E-14	1 E-14	A
RS	Resistance	0	10	Ω
BV	Reverse Breakdown Voltage	∞	5	V
IBV	Current at Reverse Breakdown Voltage	1 E-3	1 E-3	A
N	Emission Coefficient (η)	1	1	-
XT1	Saturation Current Temp. Coefficient	3	0	-
TT	Transit Time	0	10	nS
CJO	Zero Bias Junction Capacitance	0	50	pF
VJ	Junction Potential	1	1	V
M	Grading Coefficient	0.5	0.5	-
EG	Activation Energy	1.11	1.11	eV
KF	Flicker Noise Coefficient	0	0	-
AF	Flicker Noise Exponent	1	1	-
FC	Depletion Capacitance Forward Bias Coefficient	0.5	0.5	-
TNOM	Nominal Temperature	27	27	°C

1. The values of the ON macro-model components are generated by the MODPEX software from the data sheet specifications.
2. The PSPICE diode command does not have a maximum current limit (I_{MAX}) variable. The macro-model's default value for I_{ZM} is 200 mA.
3. The PSPICE diode command does not have a variable to define the temperature coefficient of the breakdown voltage. The macro-model's default value for $\Delta V_Z/\Delta T$ is 0 mV/°C.

Macro-Model Subcircuit

A Zener diode macro-model can be created using standard SPICE components. Figure 2 shows a schematic representation of the macro-model that is used to create the majority of the ON Zener SPICE models. The netlist for the MMBZ5232's macro-model is provided in Appendix I as an example. A summary of the operation of the model is given in Table 2. Further details on the macro-model are provided in references [5] and [6]. Alternative Zener diode macro-models are given in references [1]-[4].



Figure 2. Zener Diode SPICE Macro-Model

Forward Bias Region

Diode D_1 is the key component when the voltage V_D across the Zener is greater than zero. D_1's forward bias characteristics are controlled by the saturation (IS), emission coefficient (N) and series resistance (RS) variables. The forward bias current equations are listed below.

$$
\begin{aligned}
I_D &= I_F + I_L + I_R \\
&= I_{F\_D1} + \frac{V_D}{R_L} + I_{S\_D2} \qquad I_L \;\&\; I_R \ll I_{F\_D1} \\
\therefore I_D &\cong I_{F\_D1} \cong I_{S\_D1}\left[e^{\wedge}\left(\frac{V_{D1}}{\eta V_T}\right) - 1\right] \cong I_{S\_D1}\left[e^{\wedge}\left(\frac{V_{D1}}{\eta V_T}\right)\right] V_T = \frac{kT}{q} \cong 26\ \text{mV @ } 25^\circ\text{C}
\end{aligned}
$$

Leakage Region

The leakage or reverse bias before breakdown region is defined when voltage V_D is between 0 V and the breakdown voltage (V_{BR}). Currents I_F and I_R are small in comparison to I_L because diodes D_1 and D_2 are reverse biased. The leakage current can be approximated by the ratio of V_D to R_L.

$$
\begin{aligned}
I_D &= I_F + I_L + I_R \\
&= I_{S\_D1} + \frac{V_D}{R_L} - I_{S\_D2} \qquad I_{S\_D1} \;\&\; I_{S\_D2} \ll I_L \\
\therefore I_D &\cong \frac{V_D}{R_L}
\end{aligned}
$$

Breakdown Region

The reverse bias breakdown region is modeled by the devices contained in the current path of I_R. Current flows through this path when voltage V_D exceeds the voltage of EV_1 plus the forward voltage of D_2. The breakdown voltage (V_{BR}), represented by EV_1, is equal to the product of current source I_{BV} and resistor R_{BV}. The Zener voltage V_Z, specified at current I_{ZT}, is equal to the sum of the voltages of EV_1, R_Z, D_2 and D_3 as shown below.

$$
I_D \cong I_S\left[e^{\wedge}\left(\frac{V_D}{\eta V_T}\right)\right] \quad \therefore V_D \cong \eta V_T\left[\ln\left(\frac{I_D}{I_S}\right)\right]
$$

$$
V_Z = V_{BR} + V_{D2} + V_{RZ} - V_{D3}
$$

$$
\begin{aligned}
V_Z \text{ @ } I_{ZT} &= EV_1 + \eta_2 V_T \ln\left(\frac{I_{D2}}{I_{S2}}\right) + (I_{ZT}R_Z) - \eta_3 V_T \ln\left(\frac{I_{D3}}{I_{S3}}\right) \\
&= (I_{BR}R_{BV}) + \eta_2 V_T \ln\left(\frac{I_{ZT}}{I_{S2}}\right) + (I_{ZT}R_Z) - \eta_3 V_T \ln\left(\frac{I_{ZG} - I_{ZT}}{I_{S3}}\right)
\end{aligned}
$$

An equation that determines the ratio of the emission coefficients η_2 and η_3 is created by assuming that the reverse saturation currents for diodes D_2 and D_3 are equal.

$$\eta_3 = \eta_2 \ln\left(\frac{I_{ZG}-I_{ZT}}{I_{ZT}}\right)$$

The Zener impedance, or the slope of the current versus voltage curve, is equal to the derivative of the diode's voltage equation with respect to current. The impedance equation shown below is valid for low frequencies, where the junction capacitance can be neglected.

$$Z_D = \frac{\delta V_D}{\delta I_D} = \frac{\eta V_T}{I_D}$$

The impedance of the Zener, defined as Z_{ZT} and Z_{ZK}, is defined at test currents I_{ZT} and I_{ZK}, respectively. The two impedance and η_3 equations are used to calculate R_Z, η_2 and η_3. The Zener impedance in the breakdown region is approximated by neglecting the parallel connected resistor R_L and the reverse biased diode D_1 for simplicity.

$$Z_Z \cong R_Z + Z_{D2} + Z_{D3}$$

$$Z_{ZT} \cong R_Z + \frac{\eta_2 V_T}{I_{ZT}} + \frac{\eta_3 V_T}{(I_{ZG}-I_{ZT})}$$

$$Z_{ZK} \cong R_Z + \frac{\eta_2 V_T}{I_{ZK}} + \frac{\eta_3 V_T}{(I_{ZG}-I_{ZK})}$$

Current Limit Region

Diode D_3, current source I_{ZG} and resistor R_{ZG} are used to form a current limiting circuit. The current limiting mode occurs when I_{D2} increases to the value of I_{ZG}, which reverse biases D_3 and limits the current. The maximum current (I_{ZM}) has a weak dependency on the magnitude of the bias voltage and R_{ZG} is used to provide a slope to the current versus voltage curve. D_3's current can be expressed by the equations listed below.

$$I_{D3} = I_{ZG} - I_{D2} - I_{RZG}$$

$$I_{RZG} \ll I_{ZG} \;\&\; I_{D2}$$

$$\therefore I_{D3} \cong I_{ZG} - I_{D2}$$

Macro-Model Limitations

Macro-models provide an accurate SPICE representation of a Zener's current and voltage characteristics for most applications. The macro-models solve several of the limitations associated with the SPICE diode 'D' statement and the curve fit models. The accuracy of the macro-models is directly proportional to the thoroughness of the Zener's data sheet. Macro-models are a powerful analytical design tool; however, they should not be used as a replacement for hardware development tests. A summary of the limitations of the macro-models is shown in Table 3.

Table 2. Key Design Equations and Features of the Zener Diode Macro-Model

Region	Voltage Boundaries	Key Design Equations	Macro-Model Components	Comments
Forward Bias	$V_D > 0$	$I_D \cong I_{F\_D1} \cong I_{S\_D1}\left[e\,\hat{}\left(\frac{V_{D1}}{\eta V_T}\right)\right]$	D_1	• I_D is defined by the Ebers-Moll equation • D_1's RS variable models the $\Delta I/\Delta V$ slope
Leakage Region	$V_{BR} < V_D \leq 0$	$I_D \cong I_L \cong \frac{V_D}{R_L}$	R_L	• R_L's temperature coefficients model $\Delta I_L/\Delta T$
Breakdown	$V_{ZM} < V_D \leq V_{BR}$	$V_Z \cong V_{ZT} + Z_{ZT}I_D$ $V_{ZT} = V_Z @ I_{ZT}$ $Z_{ZT} \cong R_Z + RS_{D2} + RS_{D3}$	EV_1, D_2, R_Z I_{BV}, R_{BV}	• The recommended bias current for a voltage regulator application is $I_{ZM} < I_D \leq I_{ZT}$ • R_{BV}'s temperature coefficients model $\Delta V_Z/\Delta T$ • R_Z models the $\Delta I/\Delta V$ slope
Current Limit	$V_D \leq V_{ZM}$	$I_{ZM} \cong I_{ZG}$	D_3, I_{ZG}, R_{ZG}	• I_{ZM} is a function of the IC package • R_{ZG} models the $\Delta I/\Delta V$ slope

Table 3. Simulation Limits of Zener Macro-Models

Region	Key Design Parameter	Model Limitation
Forward	Forward Voltage (V_F)	• V_F is typically specified as a maximum value at a single current point in the data sheet • The simulation accuracy is enhanced if two typical current points are used by the modeling algorithm
Leakage	Leakage Current (I_L)	• I_L is modeled as a linear function of the bias voltage • I_L actually varies as an exponential function of the bias voltage
Breakdown	Zener Voltage (V_Z)	• V_Z tolerance (typ. ±5%) is not modeled • Monte-Carlo simulations can provide tolerance and worst case analysis
Current Limit	Maximum Current (I_{MAX})	• Voltage surge suppression capability beyond I_{MAX} is not modeled • Thermal self-heating produced when I_D is large is not modeled • Device overcurrent failures are not modeled. At the device's destruction point, V_Z 'collapses' or decreases to a low value, which increases the current through the device to a level that damages the device

References

1. Antognetti, P. and Massobrio, G., *"Semiconductor Device Modeling with Spice"*, McGraw-Hill, New York, 1990.
2. Deveney, M., "A Temperature Dependent SPICE Macro-Model for Zener and Avalanche Diodes", *IEEE Circuits and Systems Midwest Symposium*, May, 1991.
3. Pawlikiewicz, A. and Zack, G., "A New Macro-Model for Zeners", *IEEE Circuits and Devices Magazine, Volume 9, Issue 2*, March, 1993.
4. Sandler, S., "SPICE Subcircuit Accurately Models Zener Characteristics", *Personal Engineering*, November, 1998.
5. Wong, S. and Hu, C., "SPICE Macro-Model for the Simulation of Zener Diode I-V Characteristics", *IEEE Circuits and Devices Magazine, Volume 7, Issue 4*, July, 1991.
6. Wong, S., Hu, C. and Chan, S., "SPICE Macro-Model for the Simulation of Zener Diode Current-Voltage Characteristics", *International Journal of Electronics, Volume 71, No. 24*, August, 1991.

Appendix I: MMBZ5232B Zener Diode Macro-Model SPICE Netlist

```
******************************************************************************
*     MMBZ5232B Macro-Model SPICE Model
*     Model Generated by MODPEX
*     Copyright(c) Symmetry Design Systems
*     All Rights Reserved
*     MODEL FORMAT: PSPICE
******************************************************************************
*     node:              anode cathode
.SUBCKT mmbz5232blt1    2       1
******************************************************************************
*     Forward Region
*     D1's CJO term models the Zener's capacitance
D1 2 1 MD1
.MODEL MD1 D IS=7.58703e-07 N=3.10159 XTI=1 RS=0.856
+ CJO=1.5e-10 TT=1e-08
******************************************************************************
*     Leakage Region
*     RL models leakage current (IL)
*     MDR temp. coef. model ΔIL / ΔT
RL 1 2 MDR 1.5e+09
.MODEL MDR RES TC1=0 TC2=0
******************************************************************************
*     Reverse Breakdown Region
*     RZ models the ΔI / ΔV slope
RZ 2 3 0.862776
D2 5 4 MD2
.MODEL MD2 D IS=2.5e-12 N=0.475376 XTI=0 EG=0.1
EV1 1 5 6 0 1
*     Breakdown Voltage (VBR) = IBV x RBV
*     MDRBV temp. coef. model ΔVBR / ΔT
IBV 0 6 0.001
RBV 6 0 MDRBV 5431.8
.MODEL MDRBV RES TC1=0.000368055
******************************************************************************
*     Current Limit Region
*     Maximum current (IZM) is set to magnitude of IZG
*     RZG models the ΔI / ΔV slope
IZG 4 3 0.24
RZG 4 3 75
D3 3 4 MD3
.MODEL MD3 D IS=2.5e-12 N=0.198247 XTI=0 EG=0.1
*
.ENDS mmbz5232blt1
******************************************************************************
```

PUBLICATION ORDERING INFORMATION

LITERATURE FULFILLMENT:
Literature Distribution Center for ON Semiconductor
P.O. Box 61312, Phoenix, Arizona 85082-1312 USA
Phone: 480-829-7710 or 800-344-3860 Toll Free USA/Canada
Fax: 480-829-7709 or 800-344-3867 Toll Free USA/Canada
Email: orderlit@onsemi.com

N. American Technical Support: 800-282-9855 Toll Free USA/Canada

Japan: ON Semiconductor, Japan Customer Focus Center
2-9-1 Kamimeguro, Meguro-ku, Tokyo, Japan 153-0051
Phone: 81-3-5773-3850

ON Semiconductor Website: http://onsemi.com

Order Literature: http://www.onsemi.com/litorder

For additional information, please contact your local Sales Representative.